
ユーザーコード

• 初心者が、ユーザーコードを書くのは容易ではな
い。

• 同じ様な形状で、似た問題を扱っているユーザー
コードを自分の問題様に変えるやり方の方が良
い。

• KEKからは、様々なサンプルユーザーコードを提
供している。

• ucnai3.mor を例にして、ユーザーコードの基本構
成を説明する。



KEKから配布しているサンプルユーザーコード

• ucnai1.mor, ucnai2.mor, ucnai3.mor, ucnai4.mor
– 様々な形状のNaI(Tl) 検出器のレスポンス計算(PRESTA使用せず)

• ucnai1p.mor, ucnai2p.mor, ucnai3p.mor, ucnai4p.mor
– 様々な形状のNaI(Tl) 検出器のレスポンス計算(PRESTA使用)

• ucedep.mor
– 平板に平行電子ビームが垂直に入射した場合の、吸収エネルギーの計算

( PRESTA使用)
• ucbfsp.mor

– 点等方線源に対するγ線再生係数の計算(PRESTA使用)
• ucarray.mor, ucarrayp.mor 

– 円筒平板形状のNaI(Tl) 検出器が複数ある場合のレスポンス計算(PRESTA使用
しない場合と、使用の場合)

• ucge.mor
– Ge検出器の回りにCsI検出器がある形状でのGe検出器のレスポンス計算

(PRESTA使用)
• uccal.mor

– サンプリングカロリメータのエネルギー分解能の計算(PRESTA使用)



ucnai3.mor

• 円筒ー平板形状でのNaI検出器のレスポンス計算

Step 1

• Step 1 には、下記の項目を書く
– Over-ride EGS4 macros
– Non EGS4 macros
– COMMON とDIMENSION

– 物質データの定義
– 使用するファイルの指定– open statements



%C80

!NEWCONDITIONAL;

"------------------------------------------------------------------"

"Select random number generator: 0=RAN6 1=RANMAR "

"RANMAR is a Lagged-Fibonacci Method pseudo random number 
generator"

"devised by George Marsaglia and Arif Zaman.                       "    

"------------------------------------------------------------------"

REPLACE {$RNGEN} WITH {0}

"STEP 1.  USER-OVER-RIDE-OF-EGS-MACROS"



REPLACE {$RANDOMSET#;} WITH {

{SETR B=$RNGEN}

[IF] {COPY B}=0 [

IXX=IXX*663608941;{P1}=IXX*0.23283064E-09;IF(IXX.LT.0){P1}={P1}+1.0;

IF(IXX.EQ.IXXST) [OUTPUT;(' WARNING !'/

' Same random number will be produced.'/

' It is better to use RANNMAR random number generator.')]

]

[IF] {COPY B}=1 [

{P1}=URNDM(IXX)-URNDM(JXX); IF({P1}.LT.0.) {P1}={P1}+1.; 

URNDM(IXX) = {P1};

IXX=IXX-1; IF(IXX.EQ.0) IXX=97;

JXX=JXX-1; IF(JXX.EQ.0) JXX=97;

CRNDM=CRNDM-CDRNDM; IF(CRNDM.LT.0.) CRNDM=CRNDM+CMRNDM;

{P1}={P1}-CRNDM; IF({P1}.LT.0.) {P1}={P1}+1.;

]

}



"------------------------------------------------------------------"       

"      PLACE COMPILER DEPENDENT SUBROUTINE CALL HERE               "   

"------------------------------------------------------------------"       

" Select Fortran Compiler.                   　　 "       

" Lahey Fortran = 1                              　　　 "

" Microsoft Fortran = 2                        　　　 "

" Other PC = 3 (default)                                        "

" g77 on Windows  = 4                                           "

" Sun UNIX Workstation and Linux = 5                            "

" Other UNIX Workstation =6                                     "

"------------------------------------------------------------------"       

REPLACE {$COMPILER} WITH {3}                                    

"------------------------------------------------------------------"      



"------------------------------------------------------------------"       

"Macro to define OPEN STATEMENT for Cross-section data.            "

"                                            (UNIX or PC).      "       

"------------------------------------------------------------------"       

REPLACE {$OPEN;} WITH {                                         

{SETR B=$COMPILER}                                            

[IF] {COPY B}<5 [OPEN(12,FILE='mortjob.xse',status='old');

OPEN(6,FILE='mortjob.out',status='new');

OPEN(8,FILE='mortjob.dum',status='new');]

[ELSE]          [OPEN(12,FILE='mortjob.xsec',status='old');

OPEN(6,FILE='mortjob.output',status='new');

OPEN(8,FILE='mortjob.dummy',status='new');]

}                                                               



"COMMON to define variables to score at AUSGAB"

"DEPE:deposited energy inside the detector"

"DELTAE:energy bin width in MeV"

"SPG:Gamma spectrum, SPE:Electron spectrum, SPP:Positron spectrum"

REPLACE {;COMIN/TOTALS/;} WITH

{;COMMON/TOTALS/DEPE,DELTAE,SPG($NDET,$NEBIN),SPE($NDET,$NEBIN),

SPP($NDET,$NEBIN);}

"COMMON of print-out parameter"

REPLACE {;COMIN/LINES/;} WITH

{;COMMON/LINES/NLINES,NWRITE,NCOUNT,ILINES;}

"COMMON of geometry related parameter"

REPLACE {;COMIN/PASSIT/;} WITH

{;COMMON/PASSIT/NREG,NPLAN,NCYL,IRZ;}

[Non EGS4 macros]



PARAMETER $MXPLNS=5; "NUMBER OF PLANE"                          

PARAMETER $MXCYLS=3; "NUMBER OF CYLINDER"                       

PARAMETER $NCASES=5000; "MAXIMUN NUNBER OF CASES"               

PARAMETER $NBATCH=50;  "Number of batch" 

PARAMETER $NEBIN=50;   "Number of energy bin"

PARAMETER $NDET=1;     "Number of detector “

;COMIN/DEBUG,BOUNDS,BREMPR,CYLDTA,EDGE,ELECIN,ETALY1,LINES,MEDIA,

MISC,NTALY1,PASSIT,PLADTA,RANDOM,STACK,THRESH,TOTALS,UPHIOT,USEFUL,USER/;              

DIMENSION PH($NEBIN),PHPB($NEBIN,$NBATCH);

DIMENSION SPGPB($NDET,$NEBIN,$NBATCH),SPEPB($NDET,$NEBIN,$NBATCH),

SPPPB($NDET,$NEBIN,$NBATCH);

DIMENSION PEFPB($NBATCH),TEFPB($NBATCH);

[$PARAMETER statements]

[COMIN and DIMENSION]



[ユーザーコードで使用する物質の指定]

• 物質名は、24 文字で指定する。
• MADARR の最初の引く数は、24でなければならない。

• ２番目の引く数は、ユーザーコードで使用する物質の数。

$TYPE   MEDARR(24,3);

DATA MEDARR/$S'NAI-XRAY-IAPRIM',9* '  ',                               

$S'AL-IAPRIM',15*'  ',

$S'QUARTZ-IAPRIM',11 *'  ' /;        

テキストと異なっている



Step 2

• SUBROUTINE AUSGAB で使用する変数を、ジオメトリーに関係す
る情報と共に定義する。
– NMED: User Codeで使用する物質数

– NREG: リージョンの数
– MED(I): 物質番号
– ECUT(I), PCUT(I): EGS4カットオフエネルギー

– IEDGFL(I): 53:原子番号、Produce X rays 
0: X-ray are not produced

リージョン毎に定義す
る必要がある

KEK Extensionでは、1



KEKでの拡張(特性X線)

• 化合物、混合物の特性X線
– 従来は、1つの元素で代表(原子番号を指定)

– 拡張では、構成している全ての元素からの特
性X線を発生

• 光子エネルギー依存のsub-shell 断面積の
導入

• Auger電子の組み込み(実験での検証必要)



KEKでの拡張(Electron Impact Ionization)

• このユーザーコードでは、必要ないが、低エネル
ギー電子からの光子生成では重要

• EIIを適用するには
– COMINにEIICOMを加える

– EIIを適用したいリージョンに対して使用する断面積い
よりIMPACR(I)を設定する

– 詳細は、KEK Internal 2000-4を参照

• 散乱における偏光効果、ドップラー広がり
– 詳細は、KEK Internal 2000-4を参照



NMED=3; "NUMBER OF MEDIA"                                       

DO J=1,NMED [                                                   

DO I=1,24 [MEDIA(I,J)=MEDARR(I,J);]]                            

NPLAN=$MXPLNS;    "NUMBER OF PLANES"                            

NCYL=$MXCYLS;    "NUMBER OF CYLINDER"                           

NREG=(NPLAN-1)*NCYL+3; "NUMBER OF REGIONS (INCLUDING OUTSIDE VACUUM"        

"                       REGION)                                 "         

IRZ=NREG-3;

Ucnai3p.morのHOWFARによる円筒ー平板形状の扱い上必要



"SET MEDIUM INDEX FOR EACH REGION"                               

/MED(1),MED(NREG-1),MED(NREG)/=0;   "VACUUM REGIONS"                            

/MED(2),MED(3),MED(4),MED(7),MED(10),MED(13)/=2;

"                                  Al region"

MED(8)=1;   "NaI(Tl) detector region"

/MED(11),MED(12)/=3;   "Quartz region"

/ECUT(2),ECUT(3),ECUT(4),ECUT(7),ECUT(8),ECUT(10)/=0.561;

/ECUT(11),ECUT(12)/=0.561;

/MED(5),MED(6),MED(9)/=0;  "Vacuum region inside case"

IEDGFL(8)=53;  “53:Atomic number of I. Produce X rays"

"                          0:X rays are not produced"           



AE,AP とECUT, PCUTの違い

• AE とAP は、PEGS4 でカットオフエネルギーとして使用
• AEより小さなエネルギーの二次電子を生成する電子ある

いは陽電子の散乱及びAPより小さなエネルギーの光子
を生成する制動輻射は、連続減衰過程に含められる。

• ECUT とPCUT は、ユーザーコードで定義される。
• もし、電子や陽電子のエネルギーがECUTより小さくなった

り、光子のエネルギーがPCUTより小さくなった場合には、
その粒子の全運動エネルギーが、その場所で付与され
る。

• ECUT とPCUT が各リージョンに割り当てられない場合は、
AEとAPがECUT及びPCUTとして割り当てられる。



Step 3
• SUBROUTINE HATCH をCALLし、ユーザーコードで使

用する物質データを物質データファイルから読み込む。

• 使用する物質データに関するデータを出力する。
– 例えば、名前、密度、放射長、AE、AP、UE、UE

• 物質名や、カットオフエネルギーは、プログラムを調べる
のに有益な情報である

• 非常に多くのリージョンがある場合には、物質の割り当て
が正しく行われているかどうかを調べるのに必要なデー
タのみを出力する



CALL HATCH;

"OUTPUT VARIOUS QUANTITIES ASSOCIATED WITH THE MEDIA"           

OUTPUT;  ('1QUANTITIES ASSOCIATED WITH EACH MEDIA:',//);        

DO J=1,NMED [                                                   

OUTPUT (MEDIA(I,J),I=1,24);  (/,1X,24A1);                       

OUTPUT RHO(J),RLC(J);  (5X,' RHO=',G15.7,' G/CM**3     RLC=',   

G15.7,' CM');                                                

OUTPUT AE(J),UE(J);  (5X,' AE=',G15.7,' MEV    UE=',G15.7,' MEV');              

OUTPUT AP(J),UP(J);  (5X,' AP=',G15.7,' MEV    UP=',G15.7,' MEV');              

]                                                               

OUTPUT;(/' INFORMATION OF MEDIUM AND CUT-OFFFOR EACH REGION'//);                

DO I=1,NREG [                                                   

IF(MED(I).EQ.0) [OUTPUT I,ECUT(I),PCUT(I);                      

(' MEDIUM(',I3,')=VACUUM',18X,'ECUT=',G10.5,' MEV, PCUT=',G10.5,' MEV');        

]                                                               

ELSE [OUTPUT I,(MEDIA(II,MED(I)),II=1,24),ECUT(I),PCUT(I);      

(' MEDIUM(',I3,')=',24A1,'ECUT=',G10.5,' MEV, PCUT=',G10.5,' MEV');]            

]                                                               



Step 4

• ジオメトリーに関係するデータを割り当てる

• 平板の場合
– 個々の平板毎に、位置座標(PCOORD)と垂直ベクトル(PNORM) 

を定義する
– 意味については、後の講義で説明

• 円筒の場合
– 円筒毎に、半径 (CYRAD)と半径の自乗(CYRAD2) を定義する

• ジオメトリーのチェックのためには、定義した値を出力し
ておく方が良い



Step 5

• SUBROUTINE ECNSV1 及び NTALLY を使用する場合
には、それらを初期化する

CALL ECNSV1(0,NREG,TOTKE);"  INITIALIZE ESUM ARRAY FOR ENERGY"  

"                            CONSERVATION CALCULATION."         

"                            NREG=NUMBER OF REGIONS"            

"                            TOTKE=TOTAL KE (DUMMY VARIABLE HERE)"              

"                                  (MUST BE REAL*8)"            

CALL NTALLY(0,NREG);                                            



•AUSGABで使用する変数の初期化

•NCOUNT とILINES は、中間結果の出力を制御する変数

NCOUNT=0;  "PARTICLE HISTORY COUNTER"                           

ILINES=0;  "INITIALIZE LINE-OUTPUT COUNTER"                                     

DEPE=0.D0; "ZERO THE ENERGY DEPOSITION AT SCINTILATOR"

/PEF,TEF/=0.0;  "Zero the efficiency"

DO J=1,$NEBIN [PH(J)=0.0;]  "Zero the pulse-height"

DO ND=1,$NDET [

DO J=1,$NEBIN [

/SPG(ND,J),SPE(ND,J),SPP(ND,J)/=0.0; "Zero the spectrum"

]]



Step 6

• 入射粒子のパラメータを定義
– IQI: 粒子の種類 EI:全エネルギー、XI, YI, ZI :入射位置
– UI,VI,WI:方向余弦、IRI:入射リージョン
– WTI:入射粒子の重み (普通は =1)

• 上記に加えて下記の変数を定義する
– IDINC=-1: 入射粒子である事を示す変数
– IXXST:種乱数 (for SLAC RAN6)
– NWRITE, NLINES:中間結果の出力条件

– NBATCH:バッチに分割して計算する場合のバッチ数
– NCASES, NCASPB:ヒストリー数とバッチ当たりのヒストリー数



IQI=0;  "INCIDENT PARTICLE"                                     

EI=1.33 +ABS(IQI)*PRM;   "TOTAL ENERGY OF PARTICLE (MEV) "      

AVAILE=EI + IQI*PRM;  "AVAILABLE K.E. (MEV) (MUST BE REAL*8)"   

EKIN=AVAILE;

DELTAE=0.05;  "Energy bin of response"                          

XI=0.0; YI=0.0; ZI=0.0;  "STARTING COORDINATES (CM)"            

UI=0.0; VI=0.0; WI=1.0;  "INCIDENT DIRECTION COSINES"           

IRI=2;  "ENTRANCE REGION DEFINITION"                            

WTI=1.0;  "WEIGHT FACTOR OF UNITY"                              

IDINC=-1; "AN IDENTIFIER (LIKE IARG) TO MARK INCIDENT PARTICLES"       



IXXST=17847465;                                                 

IXX=IXXST;  "INITIALIZED RANDOM NUMBER WITH STARTING SEED"      

$RNG-INITIALIZATION;

NWRITE=10;  "NUMBER OF INCIDENT CASES TO PRINT OUT"             

NCASES=$NCASES; "MAXIMUM NUMBER OF INCIDENT CASES TO RUN"       

NBATCH=$NBATCH;  "NUMBER OF BATCH"                              

NCASPB=NCASES/NBATCH; "NUMBER OF CASES PER BATCH"               

NOFBAT=0;  "NUMBER OF BATCH FINISHED"                           

NLINES=15; "NUMBER OF LINES TO PRINT OUT"                       

Initialization if RANMAR 
random number generator 
are selected



Step 7
• 線源条件IQI, EI, XI, YI, ZI, UI, VI, WI, IRI, WTI を付加してl 

“SUBROUTINE SHOWRR” をCALLする

• ヒストリー毎に、線源の条件が異なる場合には、線源の
条件を決定するルーチンを“CALL SHOWER” の前に挿入
しておく必要がある

DO NOFBAT=1,NBATCH [ "BATCH-LOOP"

DO I=1,NCASPB ["START OF SHOWER CALL LOOP OF EACH BATCH"

IF(NCOUNT.LE.NWRITE.AND.ILINES.LE.NLINES) [                     

OUTPUT EI,XI,YI,ZI,UI,VI,WI,                                  

IQI,IRI,IDINC;   (7G15.7,3I5);                                

ILINES=ILINES+1;]                                             

CALL SHOWER(IQI,EI,XI,YI,ZI,UI,VI,WI,IRI,WTI);                  

線源のサンプリングルーチンが必要
な場合には、この場所に挿入



統計的な誤差評価

• x をモンテカルロ計算によって求める量とする

• 誤差を評価するのに便利な２つの方法がある
• MCNPで使用している方法

– 計算は N 個の“入射” 粒子について行われ、xi は、i-番目のヒスト
リーの結果であるとする

MORSE-CGで使用している方法

• 計算は N 個の“入射” 粒子について行われ、xi は、i-番目
のヒストリーの結果であるとする

• “N” ヒストリーを、それぞれ N/n ヒストリーのn 個のバッチ
に分割する

• 各バッチ毎に得られた値をxｊとする

ucnai3.mor で使用している誤差評価の方法

• MORSE-CG で使用している方法によりFSD を求める

• ヒストリー数 (NCASES) を、NBATCH このバッチに分割する
– バッチ毎のヒストリー数：NCASPB(=NCASES/NBATCH)

• 検出器中の吸収エネルギーから求められるレスポンス、ピー
ク及び全検出効率は、ヒストリー毎に計算する

"If some energy is deposited inside detector add pulse-height"

"and efficiency"

IF(DEPE.GT.0.D0) [

IE=DEPE/DELTAE+1;

IF(IE.LE.$NEBIN) [PH(IE)=PH(IE)+WTI;]

IF(DEPE.GE.EI*0.999) [PEF=PEF+WTI;]

TEF=TEF+WTI;]

DEPE=0.D0;



•平均値 は、各バッチの終わりで計算するx

"Calculate average value for this BATCH"

DO IE=1,$NEBIN [

PHPB(IE,NOFBAT)=PH(IE)/NCASPB;

PH(IE)=0.0;

]

PEFPB(NOFBAT)=PEF/NCASPB;

TEFPB(NOFBAT)=TEF/NCASPB;

/PEF,TEF/=0.0;

DO ND=1,$NDET [

DO IE=1,$NEBIN [

SPGPB(ND,IE,NOFBAT)=SPG(ND,IE)/NCASPB; "Gamma spectrum for this BATCH"

SPEPB(ND,IE,NOFBAT)=SPE(ND,IE)/NCASPB; "Electron spectrum for this BATCH"

SPPPB(ND,IE,NOFBAT)=SPP(ND,IE)/NCASPB; "Positron spectrum for this BATCH"

/SPG(ND,IE),SPE(ND,IE),SPP(ND,IE)/=0.0;

]]



Step 8
• 得られた結果を解析し、出力する

• ジオメトリーや線源条件も出力する方が良い
• もし、ユーザコードでECNSV1 や NTALLYを使用している場合

のは、これらのルーチンから得られる統計情報を出力する文も必要

"NEXT, CALL THE SUBROUTINE ECNSV1 TO WRITE-OUT THE ENERGY DEPOSITION"           

"TOTALS---TO CHECK ENERGY CONSERVATION FOR ONE THING"                     

CALL ECNSV1(1,NREG,TOTKE);                                      

CALL NTALLY(1,NREG);                                            



AUSGABの機能
• AUAGAB は、ユーザーが得たい情報を記録するサブルー

チンである
• ucnai3.morでは、NaI検出器中での沈着エネルギー、検

出器外部から、検出器に入射した各粒子のエネルギ情
報を記録している

IRL=IR(NP);   "SET LOCAL VARIABLE"                              

DPWT=WT(NP);                                                    

"KEEP TRACK OF THE ENERGY DEPOSITION---FOR CONSERVATION 
PURPOSES"               

ESUM(IQ(NP)+2,IRL,IARG+1)=ESUM(IQ(NP)+2,IRL,IARG+1)+EDEP*DPWT;  

NSUM(IQ(NP)+2,IRL,IARG+1)=NSUM(IQ(NP)+2,IRL,IARG+1) + 1;        

IF(MED(IRL).EQ.1) ["particle is inside the detector"

DEPE=DEPE+EDEP; "Add energy deposition"



IF(IRL.NE.IROLD.AND.IARG.EQ.0) [ "particle enters into detector"

IF(IQ(NP).EQ.0) ["photon"

IE=E(NP)/DELTAE+1;

IF(IE.LE.$NEBIN) [SPG(1,IE)=SPG(1,IE)+DPWT;]]

ELSEIF(IQ(NP).EQ.-1) ["Electron"

IE=(E(NP)-RM)/DELTAE;

IF(IE.LE.$NEBIN) [SPE(1,IE)=SPE(1,IE)+DPWT;]]

ELSE ["Positron"

IE=(E(NP)-RM)/DELTAE;

IF(IE.LE.$NEBIN) [SPP(1,IE)=SPP(1,IE)+DPWT;]]

] "end of entering to detector"

] "end of inside detector"

粒子が外部からNaI 検出器に入射

PMの窓からの反射を含めない場合には、
IF((IRL.NE.IROLD.AND.IARG.EQ.0).AND.(IROLD.NE.11))



HOWFARの役割

• HOWFAR は、EGS にジオメトリーに関する情報を伝えるサブルーチン
• HOWFAR は、USTEP の途中に、リージョン境界があるかどうかを調べ

る。ある場合には、

– USTEP を境界までの距離に置き換える

– IRNEW を粒子が入っていくリージョン番号に設定する
• 粒子が、ユーザーが追跡を止めたい領域（例：体系外）に達したばあい

には、DISCARD フラグを１に設定する
• より詳細は、“ジオメトリーの書き方”の講義で説明


